

REGLAS GRAMATICALES ASIGNADAS

Toledo Valentina

Leg. 53522

<programa> ::= "tablero" <identificador> "{" { <elemento> } "}" ;

<elemento> ::= <fuente> | <metrica> | <grafico> | <filtro> | <alerta> ;

<fuente> ::= "fuente" <identificador> "tipo" <tipo_fuente> "ruta" <cadena> ";" ;

<tipo_fuente> ::= "csv" | "api" | "json" ;

<metrica> ::= "metrica" <identificador> "=" <agregacion> "(" <campo> ")" ";" ;

<agregacion> ::= "suma" | "promedio" | "maximo" | "minimo" | "contar" ;

<grafico> ::= "grafico" <identificador> "{" "tipo" "=" <tipo_grafico> ";" "usar" "=" <identificador> ";" "}" ;

<tipo_grafico> ::= "barras" | "lineas" | "torta" | "tabla" ;

<filtro> ::= "filtro" <campo> <operador> <valor> ";" ;

<operador> ::= "==" | "!=" | ">" | "<" | ">=" | "<=" ;

<alerta> ::= "alerta" "si" <identificador> <operador> <valor> "entonces" <cadena> ";" ;

<campo> ::= <identificador> { "." <identificador> } ;

<valor> ::= <cadena> | <numero> | <booleano> ;

<booleano> ::= "verdadero" | "falso" ;

<identificador> ::= <letra> { <letra> | <digito> | "_" } ;

<cadena> ::= "" { <caracter> } "" ;

<numero> ::= <digito> { <digito> } ["." <digito> { <digito> }] ;

<caracter> ::= <letra> | <digito> | " " | "." | "," | ":" | "/" | "_" | "-";

<letra> ::= "a" | ... | "z" | "A" | ... | "Z" ;

<digito> ::= "0" | ... | "9" ;